Question 4 Solutions

Part (a)

```
> f1,f2,f3 := x+y+z-3, x^2+y^2+z^2-5, x^3+y^3+z^3-7;
```

$$f1, f2, f3 := x + y + z - 3, x^2 + y^2 + z^2 - 5, x^3 + y^3 + z^3 - 7$$

```
> f := x^4+y^4+z^4-9;
```

$$f := x^4 + y^4 + z^4 - 9$$

```
> with(Groebner):
> G := Basis([f1,f2,f3],grlex(x,y,z));
```

$$G := [x + y + z - 3, y^2 + y z + z^2 - 3 y - 3 z + 2, 3 z^3 - 9 z^2 + 6 z + 2]$$

```
> NormalForm(f,G,grlex(x,y,z));
```

$$0$$

So f is in the ideal <f1,f2,f3>

```
> h := x^5+y^5+z^5;
```

$$h := x^5 + y^5 + z^5$$

```
> NormalForm(h,G,grlex(x,y,z));
```

$$\frac{29}{3}$$

So h = 29/3

Part (b)

```
> f := x^3+2*x*y*z-z^2;
```

$$f := x^3 + 2 x y z - z^2$$

```
> g := x^2+y^2+z^2-1; # sphere
```

$$g := x^2 + y^2 + z^2 - 1$$

```
> L := f-lambda*g;
```

$$L := -\left(x^2 + y^2 + z^2 - 1\right) \lambda + x^3 + 2 x y z - z^2$$

```
> F := [g,diff(L,x),diff(L,y),diff(L,z)];
```

$$F := [x^2 + y^2 + z^2 - 1, -2 x \lambda + 3 x^2 + 2 y z, -2 y \lambda + 2 x z, -2 \lambda z + 2 x y - 2 z]$$

```
> G := Basis(F,plex(lambda,x,y,z)):
> G := remove(has,G,lambda);
```

$$G := [1152 z^7 - 1763 z^5 + 655 z^3 - 44 z, -1152 z^6 + 118 z^3 y + 1605 z^4 - 118 y z - 453 z^2, -6912 z^5 + 3835 y^2 z + 10751 z^3 - 3839 z, -9216 z^5 + 3835 y^3 + 3835 z^2 y + 11778 z^3 - 3835 y - 2562 z, -1152 z^5 + 3835 z^2 y - 1404 z^3 + 3835 x z + 2556 z, -19584 z^5 + 25987 z^3 + 3835 x y - 6403 z, x^2 + y^2 + z^2 - 1]$$

```
> factor(G[1]);
```

$$z (z - 1) (3 z + 2) (3 z - 2) (z + 1) \left(128 z^2 - 11\right)$$

```
> _EnvExplicit := true; # force Maple to use radicals not RootOfs
```

$$\_EnvExplicit := true$$

```
> sols := [solve(G[1]=0,z)];
```

$$sols := \left[0, 1, -1, \frac{2}{3}, -\frac{2}{3}, \frac{\sqrt{22}}{16}, -\frac{\sqrt{22}}{16}\right]$$

```
> sols := [solve(G)];
```

$$sols := \left[\{x=1, y=0, z=0\}, \{x=-1, y=0, z=0\}, \{x=0, y=1, z=0\}, \{x=0, y=-1, z=0\}, \{x=0, y=0, z=1\}, \{x=0, y=0, z=-1\}, \left\{x=-\frac{2}{3}, y=\frac{1}{3}, z=\frac{2}{3}\right\}, \left\{x=-\frac{2}{3}, y=-\frac{1}{3}, z=-\frac{2}{3}\right\}, \left\{x=-\frac{3}{8}, y=-\frac{3\sqrt{22}}{16}, z=\frac{\sqrt{22}}{16}\right\}, \left\{x=-\frac{3}{8}, y=\frac{3\sqrt{22}}{16}, z=-\frac{\sqrt{22}}{16}\right\}\right]$$

```
> for s in sols do
      eval(f,s);
  od;
```

$$1$$
$$-1$$
$$0$$
$$0$$
$$-1$$
$$-1$$
$$-\frac{28}{27}$$
$$-\frac{28}{27}$$
$$\frac{7}{128}$$
$$\frac{7}{128}$$

The maximum is 1 at

```
> sols[1];
```

$$\{x=1, y=0, z=0\}$$

Part (c)
Let C1 be the circle in the bottom left with center x1,y1, and C2 the top circle with center x2,y2 and C3 the right circle with center x3,y3 and let m be the diameter.
We need 7 equations since we have 7 unknowns.
The unit square is meant to be the outer square so x1=m/2 and y1=m/2.

```
> eqns := [x1-m/2, y1-m/2, y2+m/2-1, x3+m/2-1,
     (x1-x2)^2+(y1-y2)^2-m^2,
     (x1-x3)^2+(y1-y3)^2-m^2,
     (x2-x3)^2+(y2-y3)^2-m^2];
```

$$eqns := \left[x1 - \frac{m}{2}, y1 - \frac{m}{2}, y2 + \frac{m}{2} - 1, x3 + \frac{m}{2} - 1, (x1 - x2)^2 + (y1 - y2)^2 - m^2, (x1 - x3)^2 + (y1 - y3)^2 - m^2, (x2 - x3)^2 + (y2 - y3)^2 - m^2\right]$$

```
> G := Basis(eqns,plex(x1,x2,x3,y1,y2,y3,m)):
> factor(G[1]);
```

$$m^2 \left(m^4 - 32 m^3 + 80 m^2 - 64 m + 16\right)$$

```
> fsolve(G[1]=0);
```

0., 0., 0.5086661901, 0.7943953532, 1.349198186, 29.34774027

The right solution is the smallest +ve one m = 0.508666.
Note, the degenerate solution m=0 can be removed by using 1 - m z = 0 for a dummy z.

```
> eqns := [op(eqns),1-m*z]:
```

```
> G := Basis(eqns,plex(z,x1,x2,x3,y1,y2,y3,m)):
> G[1];
```

$$m^4 - 32 m^3 + 80 m^2 - 64 m + 16$$

Another equation is x2=y3 by symmetry. This is simpler than the quadratic equations. Let's try

```
> eqns := [x1-m/2, y1-m/2, y2+m/2-1, x3+m/2-1,
     x2-y3, #   (x1-x2)^2+(y1-y2)^2-m^2,
     (x1-x3)^2+(y1-y3)^2-m^2,
     (x2-x3)^2+(y2-y3)^2-m^2];
```

$$eqns := \left[x1 - \frac{m}{2}, y1 - \frac{m}{2}, y2 + \frac{m}{2} - 1, x3 + \frac{m}{2} - 1, x2 - y3, (x1 - x3)^2 + (y1 - y3)^2 - m^2, (x2 - x3)^2 + (y2 - y3)^2 - m^2\right]$$

```
> G := Basis(eqns,plex(z,x1,x2,x3,y1,y2,y3,m)):
> G[1];
```

$$m^4 - 32 m^3 + 80 m^2 - 64 m + 16$$

If we use the inner square so that x1 = y1 = 0 then we have

```
> eqns := [x1,y1,y2-1,x3-1,1-z*m,
        (x1-x2)^2+(y1-y2)^2-m^2,
     (x1-x3)^2+(y1-y3)^2-m^2,
     (x2-x3)^2+(y2-y3)^2-m^2];
```

$$eqns := \left[x1, y1, y2 - 1, x3 - 1, -m z + 1, (x1 - x2)^2 + (y1 - y2)^2 - m^2, (x1 - x3)^2 + (y1 - y3)^2 - m^2, (x2 - x3)^2 + (y2 - y3)^2 - m^2\right]$$

```
> G := Basis(eqns,plex(z,x1,x2,x3,y1,y2,y3,m)):
> factor(G[1]);
```

$$m^4 - 16 m^2 + 16$$

```
> fsolve(G[1]=0,m);
```

−3.863703305, −1.035276180, 1.035276180, 3.863703305

The right value is 1.035276 the smallest +ve solution